

Lógica de Programação



Funções

Roni Schanuel
13-08-2024

Funções

- Definição : Sequência de instruções executadas somente quando chamadas por um programa em execução
 - Devem executar **uma tarefa** específica
 - Um programa **pode conter diversas funções**, além da função principal **início()** , que é **obrigatória**
 - As funções executam **somente** quando chamadas à partir da função início()
 - Após a execução, o fluxo retorna ao ponto **imediatamente após** o da chamada da função
 - Uma função pode (ou não) **retornar um valor** ao bloco que a chamou
 - Uma função pode (ou não) **necessitar de um ou mais argumentos** ao ser chamada

Vamos olhar uma etapa de cada vez ...



Funções

Com as funções podemos criar pequenos trechos de programas (blocos) que podem ser reutilizados facilitando o desenvolvimento.

funcao inicio()

É a função raiz do Portugol Studio sendo obrigatório o seu uso para o funcionamento dos algoritmos. Outras funções e procedimentos, podem ser utilizados, mas a funcao inicio() sempre deverá estar presente.

Sintaxe:

```
funcao nomeFuncao(){  
}
```

Mais alguns exemplos - Repetição de código

```
programa {  
    funcao inicio(){  
        inteiro i  
        para(i=0; i<20; i++)  
            escreva("**")  
        escreva("\n")  
        escreva("Numeros entre 1 e 5\n")  
        para(i=0; i<20; i++)  
            escreva("**")  
        escreva("\n")  
        para(i=1; i<=5; i++)  
            escreva(i, "\n")  
        para(i=0; i<20; i++)  
            escreva("**")  
        escreva("\n")  
    }  
}
```

Note o código repetido. Se
tivermos que consertar,
teremos que fazer o mesmo
ajuste várias vezes



Saída:

Numeros entre 1 e 5

1

2

3

4

5

Mais alguns exemplos - Repetição de código

```
programa {  
    funcao inicio(){  
        inteiro i  
        escreve_linha()  
        escreva("Numeros entre 1 e 5\n")  
        escreve_linha()  
        para(i=1; i<=5; i++)  
            escreva(i, "\n")  
        escreve_linha()  
    }  
    funcao escreve_linha(){  
        para(i=0; i<20; i++)  
            escreva(" ")  
        escreva("\n")  
    }  
}
```

Observe a diferença ao
encapsularmos esse código
repetido em uma função =D



Saída:

Numeros entre 1 e 5

1

2

3

4

5

Procedimentos

São funções que não possuem retorno realizando apenas a execução do que está dentro do bloco do código.

```
programa
{
    funcao inicio()
    {
        mensagem()
    }

    funcao mensagem(){
        escreva("*****\n")
        escreva("**Residência 2022**\n")
        escreva("*****")
    }
}
```

chamada do procedimento

E se fizermos mais uma chamada ao procedimento?

Procedimentos com argumentos

Passando parâmetros ou argumentos em procedimentos.

```
programa
{

    funcao inicio()
    {
        mensagem(2022)
    }

    funcao mensagem(inteiro ano){
        escreva("*****\n")
        escreva("***Residência "+ano+"**\n")
        escreva("*****")
    }

}
```

chamada passando o argumento

Podemos ter um procedimento chamando outro por exemplo

```
programa
{

    funcao inicio()
    {
        entrada()
    }

    funcao entrada(){
        inteiro ano
        escreva("Digite o ano do curso:")
        leia (ano)
        mensagem(ano)
    }

    funcao mensagem(inteiro ano){
        escreva("*****\n")
        escreva("***Residência "+ano+"**\n")
        escreva("*****")
    }

}
```

Procedimentos com passagem de parâmetros por valor

Passando mais de um parâmetro em procedimentos.

```
programa
{
    funcao inicio()
    {
        inteiro a=10, b= 20
        soma(a,b)
    }

    funcao soma(inteiro x, inteiro y){
        escreva("Soma:", x + y)
    }
}
```


Função

Toda função tem um tipo de retorno. Deve ser utilizado a palavra retorne no bloco da função.

Sintaxe:

funcao **tipoDeRetorno** **nomeDaFuncao**()

Exemplo sem parâmetro:

função **cadeia** **texto**()

Exemplo com parâmetro:

função **real** **calculaMedia**(real a, real b)

Exemplo:

O algoritmo abaixo faz a leitura de duas notas, essas notas são passadas como parâmetro por valor para a função **calculaMedia** e após o cálculo e retornada a média.

```
programa
{
    funcao inicio()
    {
        real n1, n2
        escreva("Digite a nota1:")
        leia(n1)

        escreva("Digite a nota2:")
        leia(n2)

        escreva(calculaMedia(n1,n2))
    }

    funcao real calculaMedia(real a, real b){
        retorne (a + b)/2
    }
}
```

Exemplo de função com passagem de parâmetro por valor

Quando chamamos a função **incrementa** estamos passando o valor da variável **a** para a variável **i** da função **incrementa**. Quando chamamos a função o valor original da variável **a** não é alterado.

Qual valor será impresso
no exemplo abaixo?

```
programa
{
    funcao inicio()
    {
        inteiro a = 100

        incrementa(a)
        escreva("O valor é:", incrementa(a), "\n")
        escreva("O valor é:", a)
    }

    funcao inteiro incrementa(inteiro i){
        i = i + 10
        retorne i
    }
}
```

Memória	
a	100
i	110 (100+10)

Exemplo de função com passagem de parâmetro por referência


Na passagem de parâmetros por referência a variável aponta para a variável de origem. As alterações na função afetam a variável de origem. Para usar a passagem de parâmetros por referência devemos utilizar o **&** antes do nome da variável do parâmetro.

Qual valor será impresso
no exemplo abaixo?

```
programa
{
    funcao inicio()
    {
        inteiro a = 100

        incrementa(a)
        escreva("O valor é:", incrementa(a), "\n")
        escreva("O valor é:", a)
    }

    funcao inteiro incrementa(inteiro &i){
        i = i + 10
        retorne i
    }
}
```



A variável **i** aponta para o endereço de **a**

Exemplo de função com passagem de parâmetro por referência

```
programa
{
    funcao inicio()
    {
        cadeia nome
        inteiro idade

        escreva("Digite o nome:")
        leia(nome)
        escreva("Digite a idade:")
        leia(idade)
        imprimir(nome, idade)
        escreva("-----\n")
        escreva(nome, "-", idade)

    }

    funcao imprimir(cadeia n, inteiro &i){
        escreva("Nome:", n, "\n")
        escreva("Idade:", i, "\n")
        n = "Antônio"
        i = 50
    }
}
```


Exemplo de função com passagem de parâmetro por referência

```
programa
{
    funcao inicio()
    {
        inteiro x, y
        escreva("digite os valores:")
        leia(x,y)
        escreva("O resultado:",soma(x,y),"\n")
        escreva("Valor de x:",x,"\n")
        escreva("Valor de y:",y)
    }

    funcao inteiro soma(inteiro &x, inteiro &y){
        inteiro resultado
        resultado = x + y
        x = 10
        y = 20
        retorne resultado
    }
}
```

Biblioteca de funções

São funções disponibilizadas por uma linguagem para facilitar o trabalho na criação de algoritmos. No menu ajuda do Portugol temos todas as funções disponíveis.



Para usar uma função precisamos importar para o nosso algoritmo. Vamos importar a biblioteca **Util** e utilizar a função **sorteia, aguarde e caixa_alta**.

```
programa
{
    inclui biblioteca Util
    inclui biblioteca Texto
    funcao inicio()
    {
        escreva (Util.sorteia(1,100))
        Util.aguarde(3000)
        escreva(Texto.caixa_alta("\nCurso de Programação"))
    }
}
```

O que será impresso na linha em destaque abaixo?

```
escreva(Texto.numero_caracteres(Texto.caixa_alta("\nCurso de Programação")))
```

Biblioteca de funções

Exemplo:

```
programa
{
    inclui biblioteca Matematica --> mat

    funcao inicio()
    {
        real base, expoente, resultado
        escreva("Digite a base: ")
        leia(base)

        escreva("Digite o expoente:")
        leia(expoente)

        resultado = mat.potencia(base, expoente)
        escreva( base, " elevado a ", expoente, " = ", resultado)

        escreva("\nMaior número:", mat.maior_numero(base, expoente))
    }
}
```

potencia é um exemplo de uma função existente na biblioteca de funções do Portugol

Biblioteca de funções

Exemplo:

```
programa
{
    inclui biblioteca Util --> u

    funcao inicio()
    {
        inteiro vetor[10]

        para(inteiro i=0; i < 10; i++){
            vetor[i] = u.sorteia(100,1000)
        }

        para(inteiro i=0; i < 10; i++){
            escreva(i + "-" +vetor[i], "\n")
        }

    }
}
```

Utilizando a função **sorteia** da biblioteca **Util**

Biblioteca de funções

Exemplo:

```
programa
{
    inclui biblioteca Util --> util

    funcao inicio()
    {
        inteiro numero = 0 // Declara uma variável
        sorteio (numero)
        escreva("Número:")
        exibe (numero)
    }

    funcao sorteio (inteiro &n)
    {
        n = util.sorteia (1, 100)
        escreva(n, "\n")
    }

    funcao exibe (inteiro n)
    {
        escreva (n, " ")
        n = 5
    }
}
```

Utilizando a função **sorteia** da biblioteca **Util**

Exercícios

1) Criar um matriz 5x2 do tipo cadeia onde deverá ser inicializada com usuários e senhas. Na função inicio o usuário deverá ler um usuário e senha. Deverá ser criada uma função para buscar na matriz se este usuário e senha foram encontrados e retornar se achou ou não este usuário, caso o usuário for encontrado escreva uma mensagem "Bem vindo ao sistema"

2) Escrever um algoritmo que calcule o salario final de um vendedor, mostre o salário fixo, o valor de vendas e o salario final incluindo a comissão sobre as vendas no mês.

OBS: O algoritmo devera ler:

- O nome do vendedor
- Seu salário fixo
- O valor das vendas no mes
- Sua comissão sobre o valor das vendas efetuadas (em percentual)
- O sistema deverá ter um critério de saída para finalizar o programa

O SISTEMA DEVERÁ APRESENTAR O SEGUINTE RESULTADO (EXEMPLO)

```
// ----- SISTEMA DE GESTÃO DE VENDEDORES -----  
// -----
```

```
// >>> Digite o nome do vendedor: João da Silva  
// >>> Digite o salário: 1200  
// >>> Informe o valor das suas vendas deste no mês: 550  
// >>> Digite a comissão (em percentual): 10
```

```
// ----- R E S U M O -----  
// -- Nome: João da Silva  
// -- Salário Líquido: 1200.00  
// -- Valor comissão: 55.00  
// -- Salário Final (salário + comissão): 1255.00  
// -----
```

Exercícios

- 3) Crie uma função que receba um valor como parâmetro e retorne o dobro do valor e exiba na tela este valor.
- 4) Crie uma função que receba um número como parâmetro e escreva a tabuada desse número.